/*
 *  Licensed to the Apache Software Foundation (ASF) under one
 *  or more contributor license agreements.  See the NOTICE file
 *  distributed with this work for additional information
 *  regarding copyright ownership.  The ASF licenses this file
 *  to you under the Apache License, Version 2.0 (the
 *  "License"); you may not use this file except in compliance
 *  with the License.  You may obtain a copy of the License at
 *
 *    http://www.apache.org/licenses/LICENSE-2.0
 *
 *  Unless required by applicable law or agreed to in writing,
 *  software distributed under the License is distributed on an
 *  "AS IS" BASIS, WITHOUT WARRANTIES OR CONDITIONS OF ANY
 *  KIND, either express or implied.  See the License for the
 *  specific language governing permissions and limitations
 *  under the License.
 */

plugins {
    id 'signing'
    id 'org.gradlex.reproducible-builds'
}

def srcSpec = copySpec {
    from fileTree(rootProject.projectDir) {
        exclude '.circleci', '.github', '.idea',
            '**/build', '.asf.yaml', '.tm_properties',
            'gradlew', 'gradlew.bat', 'gradle/wrapper', // gradlew
            'out',             // used by Intellij IDEA
            '**/*.iml',        // used by Intellij IDEA
            '**/*.ipr',        // used by Intellij IDEA
            '**/*.iws',        // used by Intellij IDEA
            '.settings',       // used by Eclipse
            '.classpath',      // used by Eclipse
            '.gradle'          // used by Gradle
    }
}

tasks.register('distSrc', Zip) {
    archiveBaseName = 'apache-groovy-geb'
    archiveAppendix = 'src'
    into "groovy-geb-${project.version}"
    with srcSpec
}

tasks.register('createChecksums') {
    doLast {
        distSrc.outputs.files.each { file ->
            ['md5', 'sha1', 'sha256', 'sha512'].each { alg ->
                ant.checksum file: file, algorithm: alg
            }
        }
    }
}

distSrc.finalizedBy(createChecksums)

signing {
    sign distSrc
}
